
***********************************************************
******* This One’s for the Boys: 			 
******* How Gendered Political Socialization Limits Girls’ Political Ambition and Interest
******* August 2021 - data collected by authors  
******* Angie Bos Jill Greenlee Mirya Holman Zoe Oxley J Celeste Lay  
***********************************************************

**** if not already installed, please install
* scheme mrc
* coefplot 

clear
set more off
set scheme mrc

**** set working directory here **** 

use  "bosetal_data.dta"

*** setting global controls *** 

global demographics age white black hispanic 
global demographics2 white black hispanic 
global race white black hispanic 


***** boys vs girls mean level of interest for four professions 

eststo clear 

mixed doctor gender_dummy $demographics  || location: || classroom:
eststo doc: margins, at(gender_dummy=(0,1))   atmeans post  	
mixed teacher gender_dummy $demographics  || location: || classroom:
eststo teach:  margins, at(gender_dummy=(0,1))   atmeans post
mixed business gender_dummy $demographics  || location: || classroom:
eststo business:  margins, at(gender_dummy=(0,1))   atmeans post
mixed police gender_dummy $demographics  || location: || classroom:
eststo police:  margins, at(gender_dummy=(0,1))   atmeans post

	coefplot (doc, label (Doctor)) (teach, label(Teacher)) (business, label(Business person)  bcolor(olive_teal)) (police, label(Police)), ///
	ytitle ("% Selecting Career") ///
	title ("") ///
	ylab(0 "0%" .2 "20%" .4 "40%" .6 "60%" .8 "80%" 1 "100%", labsize(small)) ///
	xlab(1 "Boys" 2 "Girls", labsize(small)) ///
	vertical recast(bar) barwidth(0.2) fcolor(*.5) ///
	ciopts(recast(rcap)) citop citype(normal) format(%9.2f) ///
	legend (rows(1) pos(6) size(small))
	graph save Graph "fig1_gender_career.gph", replace
	graph export "fig1_gender_career.pdf", as(pdf) replace
	
	
	 
	# delimit ;
		esttab using appD1b_gender_career.rtf, nogap se b(%9.3f) nostar
			 label nonumbers
			 mlabel("Doctor" "Teacher" "Business" "Police")
			//addnote("Dependent variable")//
			compress replace;
		#delimit cr

	
	
	
eststo clear 
eststo doctor: mixed doctor gender_dummy $demographics  know_scale || location: || classroom:
eststo teacher: mixed teacher gender_dummy $demographics  know_scale || location: || classroom:
eststo business: mixed business gender_dummy $demographics  know_scale || location: || classroom:
eststo police:  mixed police gender_dummy $demographics  know_scale || location: || classroom:
# delimit ;
		esttab using appD1a_gender_career.rtf, nogap se b(%9.3f) starlevels(^ .10 * .05 ** .01 *** .001) 
			 label nonumbers
			//addnote("Dependent variable")//
			compress replace;
		#delimit cr
	 
eststo clear 
 mixed doctor $demographics know_scale if gender_dummy == 0 || location: || classroom:
	eststo doc_b:  margins, at(age=(6(1)12)) level(84)   atmeans post  
 mixed doctor age white  know_scale if gender_dummy == 1 || location: || classroom:
	eststo doc_g:   margins, at(age=(6(1)12)) level(84)    atmeans post 

	
coefplot (doc_b, label (Boys)) (doc_g, label (Girls) lpattern(dash)), ///
	ytitle ("% choosing career", size(large))   title("Doctor") legend (row(1) pos(6)) ///
	xtitle ("Age") ///
	xlab(6 7 8 9 10 11 12, labsize(small)) ///
	ylab(0 "0%" .2 "20%" .4 "40%" .6 "60%" .8 "80%" 1 "100%", labsize(small)) ///
	level (84) ///
	at recast(line) lwidth(*2) ciopts(recast(rarea) fintensity(30)) 
	graph save doctor.gph, replace
	
eststo clear 
 mixed teacher   $demographics know_scale if gender_dummy == 0 || location: || classroom:
	eststo teach_b: margins, at(age=(6(1)12))  atmeans post 
 mixed teacher  $demographics  know_scale if gender_dummy == 1 || location: || classroom:
	eststo teach_g: margins, at(age=(6(1)12))   atmeans post 
	
coefplot (teach_b, label (Boys)) (teach_g, label (Girls) lpattern(dash)), ///
	ytitle ("")   title("Teacher") legend (row(1) pos(6)) ///
	xtitle ("Age") ///
	xlab(6 7 8 9 10 11 12, labsize(small)) ///
	ylab(0 "0%" .2 "20%" .4 "40%" .6 "60%" .8 "80%" 1 "100%", labsize(small)) ///
	level (84) ///
	at recast(line) lwidth(*2) ciopts(recast(rarea) fintensity(30)) 
	graph save teacher.gph, replace

grc1leg doctor.gph teacher.gph, ycommon ///
	col(2) graphregion(margin(none)) ysize(3.5) 	
graph save "fig2_teacher_doc.gph", replace
graph export "fig2_teacher_doc.pdf", replace 	

eststo clear 

 eststo doc_b: mixed doctor $demographics know_scale if gender_dummy == 0 || location: || classroom:
 eststo doc_g: mixed doctor $demographics know_scale if gender_dummy == 1 || location: || classroom:
 eststo teach_b: mixed teacher   $demographics know_scale if gender_dummy == 0 || location: || classroom:
 eststo teach_g: mixed teacher  $demographics  know_scale if gender_dummy == 1 || location: || classroom:

		# delimit ;
		esttab  ///
		using appD2_fem_age_career.rtf, nogap se b(%9.3f) starlevels(^ .10 * .05 ** .01 *** .001) 
			title("Gendered socialization, feminine careers" )
			mtitle("Doctor - Boys" "Doctor - Girls" "Teacher - Boys" "Teacher - Girls")
			 label nonumbers
			//addnote("Dependent variable")//
			compress replace;
		#delimit cr

	
	
eststo clear 
 mixed business  i.gender_dummy##c.age $demographics know_scale  || location: || classroom:
	eststo bus_b: margins, at(age=(6(1)12)  gender_dummy =(0))  atmeans post 
 mixed business i.gender_dummy##c.age  $demographics  know_scale  || location: || classroom:
	eststo bus_g: margins, at(age=(6(1)12) gender_dummy =(1))   atmeans post 
	
coefplot (bus_b, label (Boys)) (bus_g, label (Girls) lpattern(dash)), ///
	ytitle ("% choosing career", size(large))   title("Business") legend (row(1) pos(6)) ///
	xtitle ("Age") ///
	xlab(6 7 8 9 10 11 12, labsize(small)) ///
	ylab(0 "0%" .2 "20%" .4 "40%" .6 "60%" .8 "80%" 1 "100%", labsize(small)) ///
	level (84) ///
	at recast(line) lwidth(*2) ciopts(recast(rarea) fintensity(30)) 
	graph save business.gph, replace

		
eststo clear 
 mixed police  i.gender_dummy##c.age $demographics know_scale  || location: || classroom:
	eststo pol_b: margins, at(age=(6(1)12)  gender_dummy =(0))  atmeans post 
 mixed police i.gender_dummy##c.age  $demographics  know_scale  || location: || classroom:
	eststo pol_g: margins, at(age=(6(1)12) gender_dummy =(1))   atmeans post 
	
coefplot (pol_b, label (Boys)) (pol_g, label (Girls) lpattern(dash)), ///
	ytitle ("")   title("Police") legend (row(1) pos(6)) ///
	xtitle ("Age") ///
	xlab(6 7 8 9 10 11 12, labsize(small)) ///
	ylab(0 "0%" .2 "20%" .4 "40%" .6 "60%" .8 "80%" 1 "100%", labsize(small)) ///
	level (84) ///
	at recast(line) lwidth(*2) ciopts(recast(rarea) fintensity(30)) 
	graph save police.gph, replace

	
grc1leg business.gph police.gph, ycommon ///
col(2) graphregion(margin(none)) ysize(3.5) 	
graph save "fig3_business_police.gph", replace
graph export "fig3_business_police.pdf", replace 
	

eststo clear 

 eststo doc_b: mixed business $demographics know_scale if gender_dummy == 0 || location: || classroom:
 eststo doc_g: mixed business $demographics know_scale if gender_dummy == 1 || location: || classroom:
 eststo teach_b: mixed police   $demographics know_scale if gender_dummy == 0 || location: || classroom:
 eststo teach_g: mixed police  $demographics  know_scale if gender_dummy == 1 || location: || classroom:

		# delimit ;
		esttab  ///
		using appD3_masc_age_career.rtf, nogap se b(%9.3f) starlevels(^ .10 * .05 ** .01 *** .001) 
			title("Gendered socialization, masculine careers" )
			mtitle("Business - Boys" "Business - Girls" "Police - Boys" "Police - Girls")
			 label nonumbers
			//addnote("Dependent variable")//
			compress replace;
		#delimit cr

		
		*** additional tests of gender differences at age points **** 
	
eststo clear 	

 
 eststo doc_8: mixed doctor $demographics  know_scale gender_dummy if young_kids == 1 || location: || classroom:
 eststo doc_10: mixed doctor $demographics  know_scale gender_dummy if older_kids == 1 || location: || classroom: 
 
 eststo teach_8: mixed teacher  $demographics  know_scale gender_dummy if young_kids == 1  || location: || classroom:
 eststo teach_10: mixed teacher  $demographics  know_scale gender_dummy if older_kids == 1|| location: || classroom:
 
 eststo business_8: mixed business  $demographics  know_scale gender_dummy if young_kids == 1|| location: || classroom:
 eststo business_10: mixed business  $demographics  know_scale gender_dummy if older_kids == 1 || location: || classroom:
 
 eststo police_8: mixed police  $demographics  know_scale gender_dummy if young_kids == 1 || location: || classroom:
 eststo police_10: mixed police  $demographics  know_scale gender_dummy if older_kids == 1 || location: || classroom:
 
 	# delimit ;
		esttab  ///
		using appD4_career_age.rtf, nogap se b(%9.3f) starlevels(* .05 ** .01 *** .001)  
			title("Gendered socialization by age" )
			mtitle("Doctor age 6-7" "Doctor age 11-12" "Teacher age 6-7" "Teacher age 11-12" "Business age 6-7" "Business age 11-12"	"Police age 6-7"	"Police age 11-12")
			keep(gender_dummy)
			 label nonumbers
			//addnote("Dependent variable")//
			compress replace;
		#delimit cr
	
****** known political leader and political activity ***** 

eststo clear 


 logit knownleader  i.gender_dummy $demographics know_scale i.location
	eststo known: margins, at(age=(6(1)12))  atmeans post 
 logit politicaldapl   i.gender_dummy $demographics  know_scale i.location
	eststo pol_act: margins, at(age=(6(1)12))   atmeans post 
	
coefplot (known, label (Known leader)) (pol_act, label (Political activity) lpattern(dash)), ///
	ytitle ("% of DAPLs")   title("") legend (row(1) pos(6)) ///
	xtitle ("Age") ///
	xlab(6 7 8 9 10 11 12, labsize(small)) ///
	ylab(0 "0%" .25 "25%" .5 "50%" .75 "75%" 1 "100%", labsize(small)) ///
	level (84) ///
	at recast(line) lwidth(*2) ciopts(recast(rarea) fintensity(30)) 
	graph save fig4_political_dapl.gph, replace
	graph export "fig4_political_dapl.pdf", replace 	

eststo clear 

 eststo known: mixed knownleader  i.gender_dummy $demographics know_scale || location: || classroom:
 eststo pol_act: mixed politicaldapl   i.gender_dummy $demographics   know_scale || location: || classroom:
	
			# delimit ;
		esttab  ///
		using appD5_known_leader.rtf, nogap se b(%9.3f) starlevels(^ .10 * .05 ** .01 *** .001)  
			title("Drawing know leader and political material in DAPL " )
			 label nonumbers
			//addnote("Dependent variable")//
			compress replace;
		#delimit cr

****** exposure by age **** 

eststo clear 
 

 mixed exposure_scale  i.gender_dummy $demographics know_scale || location: || classroom:
	eststo exposure: margins, at(age=(6(1)12))  atmeans post 
	
		
coefplot (exposure), ///
	ytitle ("How often exposed to political materials")   title("") legend (row(1) pos(6) size(small)) ///
	xtitle ("Age") ///
	xlab(6 7 8 9 10 11 12, labsize(small)) ///
	ylab(, labsize(small)) ///
	level (84) ///
	ylab(1 "Never (1)" 2 "Rarely (2)" 3 "Sometimes (3)" 4 "Often (4)") ///
	legend (off) ///
	at recast(line) lwidth(*2) ciopts(recast(rarea) fintensity(30)) 
	graph save fig5_exposure_age.gph, replace
	graph export "fig5_exposure_age.pdf", replace 	
	
eststo clear 

 eststo exposure: mixed exposure_scale  gender_dummy $demographics know_scale || location: || classroom:

			# delimit ;
		esttab  ///
		using appD6_exposure.rtf, nogap se b(%9.3f) starlevels(^ .10 * .05 ** .01 *** .001) 
			title("Exposure to political materials" )
			 label nonumbers
			//addnote("Dependent variable")//
			compress replace;
		#delimit cr
		
		
eststo clear 
		
**** children more often view political leaders to be men *** 

**** the below results are exported to excel to generate figure 6 **** 

tab gender_dapl 

sum dapl_men dapl_woman

oneway dapl_men  gender_dummy, tab
oneway dapl_woman  gender_dummy, tab



***** man in dapl by age **** 

eststo clear 
logit dapl_men i.gender_dummy##c.age  $demographics  know_scale
	eststo dapl_gender_w: margins, at(age=(6(1)12) gender_dummy =(1))   atmeans post 
logit dapl_men i.gender_dummy##c.age   $demographics  know_scale
	eststo dapl_gender_m: margins, at(age=(6(1)12) gender_dummy =(0))   atmeans post 

	coefplot (dapl_gender_m, label(Boys)) (dapl_gender_w, label(Girls) lpattern(dash)) , ///
	xtitle ("Age") ///
	ytitle ("Probability of drawing man in DAPL") ///
	title ("") legend(rows(1) pos(6)) ///
	at recast(line) lwidth(*2) ciopts(recast(rarea) fintensity(30))  ///
	ylab(0 "0%" .25 "25%" .5 "50%" .75 "75%" 1 "100%", labsize(small)) ///
	xlab(6 "6" 7 "7" 8 "8" 9 "9" 10 "10" 11 "11" 12 "12", labsize(small)) 
	graph save fig7_men_dapl.gph, replace
	graph export fig7_men_dapl.pdf, replace 
	
	
eststo clear 
	
eststo dapl_man: mixed dapl_men i.gender_dummy##c.age  $demographics  know_scale || location: || classroom:
eststo dapl_woman: mixed  dapl_woman i.gender_dummy##c.age  $demographics  know_scale || location: || classroom:
			# delimit ;
		esttab  ///
		using appD7_dapl_man.rtf, nogap se b(%9.3f) starlevels(^ .10 * .05 ** .01 *** .001) r2(%9.3f) 
			title("% drawing man or women in DAPL" )
			 label nonumbers
			 mtitle("Drew man in DAPL" "Drew woman in DAPL")
			//addnote("Dependent variable")//
			compress replace;
		#delimit cr
		
**** traits of DAPL **** 


**** children view political leadership as masculine *** 
tab masc_trait 
tab fem_trait
sum masc_trait fem_trait

oneway masc_trait  dapl_men, tab
oneway masc_trait  dapl_woman, tab
oneway fem_trait  dapl_men, tab
oneway fem_trait  dapl_woman, tab


oneway masc_trait  gender_dummy, tab
oneway fem_trait  gender_dummy, tab

graph bar (mean) dapl_men (mean) dapl_woman, over(gender_dummy, ///
relabel(1 "Boys" 2 "Girls")) ytitle(% drawing) title(Leader gender) ///
ylab(0 "0" .15 "15%" .3 "30%" .45 "45%" .60 "60%" .75 "75%") ///
legend(order(1 "Drew man" 2 "Drew woman" ))
	graph save dapl_gender.gph, replace

graph bar (mean) masc_trait (mean) fem_trait, over(gender_dummy, ///
relabel(1 "Boys" 2 "Girls")) ytitle(% drawing) title(Leader traits) ///
ylab(0 "0" .15 "15%" .3 "30%" .45 "45%" .60 "60%" .75 "75%") ///
legend(order(1 "Masculine" 2 "Feminine" ))
	graph save dapl_trait.gph, replace

graph bar (mean) man_masc (mean) fem_masc, over(gender_dummy, relabel(1 "Boys" 2 "Girls")) ///
ytitle(% drawing) title(Leader gender x masculine traits) ///
ylab(0 "0%" .1 "10%" .20 "20%" .30 "30%")  ///
legend(order(1 "Man x Masculine" 2 "Woman x Masculine" ))
	graph save dapl_masc.gph, replace

graph bar (mean) man_fem (mean) fem_fem, over(gender_dummy, relabel(1 "Boys" 2 "Girls")) ///
ytitle(% drawing) title(Leader gender x feminine traits) ///
ylab(0 "0%" .1 "10%" .20 "20%" .30 "30%") ///
legend(order(1 "Man x Feminine"  2 "Woman x Feminine"  ))
	graph save dapl_fem.gph, replace


	graph combine dapl_gender.gph dapl_trait.gph , scheme(s1mono) ycommon ///
	col(2) graphregion(margin(none)) ysize(3.5) 
	graph save dapl_gender_trait.gph, replace
	
	
	graph combine dapl_masc.gph dapl_fem.gph , scheme(s1mono) ycommon ///
	col(2) graphregion(margin(none)) ysize(3.5) 
	graph save dapl_genderxtrait.gph, replace
	
	graph combine dapl_gender_trait.gph  dapl_genderxtrait.gph , scheme(s1mono) ///
	col(1) graphregion(margin(none)) ysize(3.5) 
	graph save fig8_dapl_bar.gph, replace
	graph export fig8_dapl_bar.pdf, replace 
	

oneway man_masc gender_dummy, tab 
oneway fem_masc gender_dummy, tab 
oneway man_fem gender_dummy, tab 
oneway fem_fem gender_dummy, tab 
		

eststo clear 
mixed man_masc  $demographics  know_scale any_traits if gender_dummy == 0  || location: || classroom: 
eststo dapl_gender_m: margins, at(age=(6(1)12))  atmeans post
mixed man_masc  $demographics  know_scale any_traits if gender_dummy == 1  || location: || classroom: 
eststo dapl_gender_w:  margins, at(age=(6(1)12))  atmeans post
	coefplot  (dapl_gender_m, label(Boys)) (dapl_gender_w, label(Girls)  lpattern(dash)), ///
	xtitle ("Age") ///
	ytitle ("Man x Masculine") ///
	title ("") legend(rows(1) pos(6)) ///
	at recast(line) lwidth(*2) ciopts(recast(rarea) fintensity(30))  ///
	levels(84) ///
	xlab(6 "6" 7 "7" 8 "8" 9 "9" 10 "10" 11 "11" 12 "12", labsize(small)) ///
	ylab(0 "0%" .1 "10%" .2 "20%" .3 "30%" .4 "40%") 
	graph save man_masc_dapl.gph, replace
	

eststo clear 
mixed man_fem  $demographics  know_scale  any_traits if gender_dummy == 0 || location: || classroom: 
eststo dapl_gender_m: margins, at(age=(6(1)12))  atmeans post
mixed man_fem  $demographics  know_scale  any_traits if gender_dummy == 1 || location: || classroom: 
eststo dapl_gender_w:  margins, at(age=(6(1)12))  atmeans post
	coefplot  (dapl_gender_m, label(Boys)) (dapl_gender_w, label(Girls)  lpattern(dash)), ///
	xtitle ("Age") ///
	ytitle ("Man x feminine") ///
	title ("") legend(rows(1) pos(6)) ///
	at recast(line) lwidth(*2) ciopts(recast(rarea) fintensity(30))  ///
	levels(84) ///
	xlab(6 "6" 7 "7" 8 "8" 9 "9" 10 "10" 11 "11" 12 "12", labsize(small)) ///
	ylab(0 "0%" .1 "10%" .2 "20%" .3 "30%" .4 "40%") 
	graph save man_fem_dapl.gph, replace

	
eststo clear 
mixed fem_masc  $demographics  know_scale  any_traits if gender_dummy == 0  || location: || classroom: 
eststo dapl_gender_m: margins, at(age=(6(1)12))  atmeans post
mixed fem_masc  $demographics  know_scale  any_traits if gender_dummy == 1  || location: || classroom: 
eststo dapl_gender_w:  margins, at(age=(6(1)12))  atmeans post
	coefplot (dapl_gender_m, label(Boys)) (dapl_gender_w, label(Girls) lpattern(dash)) , ///
	xtitle ("Age") ///
	ytitle ("Woman x masculine") ///
	title ("") legend(rows(1) pos(6)) ///
	at recast(line) lwidth(*2) ciopts(recast(rarea) fintensity(30))  ///
	levels(84) ///
	xlab(6 "6" 7 "7" 8 "8" 9 "9" 10 "10" 11 "11" 12 "12", labsize(small)) ///
	ylab(0 "0%" .05 "5%" .1 "10%" .15 "15%" .2 "20%") 
	graph save fem_masc_dapl.gph, replace
	
eststo clear 
mixed fem_fem  $demographics  know_scale any_traits if gender_dummy == 0  || location: || classroom: 
eststo dapl_gender_m: margins, at(age=(6(1)12))  atmeans post
mixed fem_fem  $demographics  know_scale  any_traits if gender_dummy == 1  || location: || classroom: 
eststo dapl_gender_w:  margins, at(age=(6(1)12))  atmeans post
	coefplot  (dapl_gender_m, label(Boys)) (dapl_gender_w, label(Girls) lpattern(dash)), ///
	xtitle ("Age") ///
	ytitle ("Woman x feminine") ///
	title ("") legend(rows(1) pos(6)) ///
	at recast(line) lwidth(*2) ciopts(recast(rarea) fintensity(30))  ///
	levels(84) ///
	xlab(6 "6" 7 "7" 8 "8" 9 "9" 10 "10" 11 "11" 12 "12", labsize(small)) ///
	ylab(0 "0%" .05 "5%" .1 "10%" .15 "15%" .2 "20%") 
	graph save fem_fem_dapl.gph, replace
	
	
	
	graph combine man_masc_dapl.gph man_fem_dapl.gph fem_masc_dapl.gph fem_fem_dapl.gph, scheme(s1mono)  ///
	col(2) graphregion(margin(none)) ysize(3.5) 
	graph save fig9_traits_dapl.gph, replace
	graph export fig9_traits_dapl.pdf, replace 
	
	eststo clear 
eststo a: mixed man_masc  $demographics  know_scale any_traits if gender_dummy == 0  || location: || classroom: 
eststo b: mixed man_masc  $demographics  know_scale any_traits if gender_dummy == 1  || location: || classroom: 
eststo c: mixed man_fem  $demographics  know_scale  any_traits if gender_dummy == 0 || location: || classroom: 
eststo d: mixed man_fem  $demographics  know_scale  any_traits if gender_dummy == 1 || location: || classroom:


		# delimit ;
		esttab  ///
		using appD8_dapl_man_traits.rtf, nogap se b(%9.3f) starlevels(^ .10 * .05 ** .01 *** .001) r2(%9.3f) 
			title("Drawing men & gendered traits" )
			 label nonumbers
			//addnote("Dependent variable")//
			compress replace;
		#delimit cr
	
	eststo clear 
 
eststo a: mixed fem_masc  $demographics  know_scale  any_traits if gender_dummy == 0  || location: || classroom: 
eststo b: mixed fem_masc  $demographics  know_scale  any_traits if gender_dummy == 1  || location: || classroom: 
eststo c: mixed fem_fem  $demographics  know_scale any_traits if gender_dummy == 0  || location: || classroom: 
eststo d: mixed fem_fem  $demographics  know_scale  any_traits if gender_dummy == 1  || location: || classroom: 
	
	
		# delimit ;
		esttab  ///
		using appD9_dapl_woman_traits.rtf, nogap se b(%9.3f) starlevels(^ .10 * .05 ** .01 *** .001) r2(%9.3f) 
			title("Drawing women & gendered traits" )
			 label nonumbers
			//addnote("Dependent variable")//
			compress replace;
		#delimit cr
	
	
**** assessing overall interest by gender *** 


eststo clear 

estpost  summarize interest_scale interest_job interest_poljob per_poljobs 
esttab  using summary_interest_ambition.rtf, cells("mean") b(%9.3f)  label nonumber replace ///
addnote("Key variables of interest.")

eststo clear
sort gender_dummy 
eststo girls:  estpost sum interest_scale interest_job interest_poljob per_poljobs  if gender_dummy == 1
eststo boys:  estpost sum   interest_scale interest_job interest_poljob per_poljobs   if gender_dummy == 0
esttab boys girls ///
using tab1_interest_ambition_gender.rtf, nogap cells("mean" "sd")  b(%9.2f)  ///
nonumbers mtitles("Boys" "Girls")  label nonumber replace ///
addnote("Key variables of interest.")

**** below results are found in table 1 **** 

ttest interest_scale, by(gender_dummy) unequal
ttest interest_poljob, by(gender_dummy) unequal
ttest per_poljobs, by(gender_dummy) unequal



**** looking by age **** 


eststo clear 

mixed interest_scale $demographics  if gender_dummy == 0  || location: || classroom: 
	eststo interest_m: margins, at(age=(6(1)12))   atmeans post 
mixed interest_scale $demographics know_scale if gender_dummy == 1 || location: || classroom: 
	eststo interest_w: margins, at(age=(6(1)12))   atmeans post 
	
	
mixed per_poljobs   $demographics  if gender_dummy == 0  || location: || classroom:
eststo politics_b: margins, at(age=(6(1)12))   atmeans post 
mixed per_poljobs  $demographics  if gender_dummy == 1 || location: || classroom:
eststo politics_g: margins, at(age=(6(1)12))   atmeans post 
	
			
	coefplot  (interest_m, label(Boys)) (interest_w, label(Girls) lpattern(dash)), ///
	xtitle ("Age") ///
	ytitle ("Interest in Politics") ///
	title ("") legend(rows(1) pos(6)) ///
	level (84) ///
	vertical recast(bar) barwidth(0.3) fcolor(*.5) ///
	ylab(2.2 2.4 2.6 2.8 3.0, labsize(small)) ///
	noci ///
	xlab(1 "6" 2 "7" 3 "8" 4 "9" 5 "10" 6 "11" 7 "12", labsize(small))
	graph save interest_gender_age.gph, replace
	
	
	coefplot  (politics_b, label(Boys)) (politics_g, label(Girls) lpattern(dash)), ///
	xtitle ("Age") ///
	ytitle ("% Political jobs") ///
	title ("") legend(rows(1) pos(6)) ///
	level (84) ///
	ylab(0 "0%" .1 "10%" .2 "20%" .3 "30%" .4 "40%" .5 "50%", labsize(small)) ///
	vertical recast(bar) barwidth(0.3) fcolor(*.5) ///
	noci ///
	xlab(1 "6" 2 "7" 3 "8" 4 "9" 5 "10" 6 "11" 7 "12", labsize(small))
	graph save political_job_gender_age.gph, replace
	
	grc1leg "interest_gender_age.gph" "political_job_gender_age.gph", col (1) ///
	graphregion(margin(none)) ysize(3.5) 
	graph save fig10_interest_jobs_gender_noci.gph, replace
	graph export fig10_interest_jobs_gender_noci.pdf, replace 
	
	
		# delimit ;
		esttab  ///
		using appD11_pol_interest_ambition.rtf, nogap nostar wide
			title("Political interest and ambition by age" )
			mtitle("Interest Boys" "Interest Girls" "Ambition Boys" "Ambition Girls")
			 label nonumbers
			//addnote("Dependent variable")//
			compress replace;
		#delimit cr
	
eststo clear 

eststo interest_b: mixed interest_scale $demographics know_scale if gender_dummy == 0  || location: || classroom: 
eststo interest_g: mixed interest_scale $demographics know_scale if gender_dummy == 1 || location: || classroom: 
eststo ambition_b: mixed per_poljobs   $demographics know_scale if gender_dummy == 0  || location: || classroom:
eststo ambition_g: mixed per_poljobs  $demographics know_scale if gender_dummy == 1 || location: || classroom:
	
			
		# delimit ;
		esttab  ///
		using appD10_pol_interest_ambition.rtf, nogap se b(%9.3f) starlevels(^ .10 * .05 ** .01 *** .001) 
			mtitle("Interest Boys" "Interest Girls" "Ambition boys" "Ambition girls") 
			 label nonumbers
			//addnote("Dependent variable")//
			compress replace;
		#delimit cr
		
		
*** who are known leaders - table D12 **** 

eststo clear 
tab knownleader_2_TEXT 

**** location -  table D13 *** 
eststo clear 

oneway dapl_men location_region if gender_dummy == 1, tab
oneway dapl_men location_region if gender_dummy == 0, tab
oneway interest_scale location_region if gender_dummy == 1, tab
oneway interest_scale location_region if gender_dummy == 0, tab

